

teffects psmatch — Propensity-score matching

[Syntax](#)[Remarks and examples](#)[Also see](#)[Menu](#)[Stored results](#)[Description](#)[Methods and formulas](#)[Options](#)[References](#)

Syntax

```
teffects psmatch (ovar) (tvar tmvarlist [, tmodel]) [if] [in] [weight]
    [, stat options]
```

<i>tmodel</i>	Description
---------------	-------------

Model

logit	logistic treatment model; the default
probit	probit treatment model
hetprobit(<i>varlist</i>)	heteroskedastic probit treatment model

tmodel specifies the model for the treatment variable.

<i>stat</i>	Description
-------------	-------------

Stat

ate	estimate average treatment effect in population; the default
atet	estimate average treatment effect on the treated

2 teffects psmatch — Propensity-score matching

<i>options</i>	Description
Model	
<code>nneighbor(#)</code>	specify number of matches per observation; default is <code>nneighbor(1)</code>
SE/Robust	
<code>vce(vcetype)</code>	<code>vcetype</code> may be <code>vce(robust [, nn(#)])</code> ; use robust Abadie–Imbens standard errors with # matches <code>vce(iid)</code> ; use default Abadie–Imbens standard errors
Reporting	
<code>level(#)</code>	set confidence level; default is <code>level(95)</code>
<code>display_options</code>	control column formats, row spacing, line width, display of omitted variables and base and empty cells, and factor-variable labeling
Advanced	
<code>caliper(#)</code>	specify the maximum distance for which two observations are potential neighbors
<code>pstolerance(#)</code>	set tolerance for in overlap assumption
<code>osample(newvar)</code>	<code>newvar</code> identifies observations that violate the overlap assumption
<code>control(# label)</code>	specify the level of <code>tvar</code> that is the control
<code>tlevel(# label)</code>	specify the level of <code>tvar</code> that is the treatment
<code>generate(stub)</code>	generate variables containing the observation numbers of the nearest neighbors
<code>coeflegend</code>	display legend instead of statistics

`tvar` must contain integer values representing the treatment levels.

`tmvarlist` may contain factor variables; see [U] 11.4.3 Factor variables.

`by` and `statsby` are allowed; see [U] 11.1.10 Prefix commands.

`fweights` are allowed; see [U] 11.1.6 weight.

`coeflegend` does not appear in the dialog box.

See [U] 20 Estimation and postestimation commands for more capabilities of estimation commands.

Menu

Statistics > Treatment effects > Matching estimators > Propensity-score matching

Description

`teffects psmatch` estimates treatment effects from observational data by propensity-score matching. PSM imputes the missing potential outcome for each subject by using an average of the outcomes of similar subjects that receive the other treatment level. Similarity between subjects is based on estimated treatment probabilities, known as propensity scores. The average treatment effect (ATE) is computed by taking the average of the difference between the observed and potential outcomes for each subject.

Options

Model

`nneighbor(#)` specifies the number of matches per observation. The default is `nneighbor(1)`. Each individual is matched with at least the specified number of individuals from the other treatment level. `nneighbor()` must specify an integer greater than or equal to 1 but no larger than the number of observations in the smallest group.

Stat

`stat` is one of two statistics: `ate` or `atet`. `ate` is the default.

`ate` specifies that the average treatment effect be estimated.

`atet` specifies that the average treatment effect on the treated be estimated.

SE/Robust

`vce(vcetype)` specifies the standard errors that are reported. By default, `teffects psmatch` uses two matches in estimating the robust standard errors.

`vce(robust [, nn(#)])` specifies that robust standard errors be reported and that the requested number of matches be used optionally.

`vce(iid)` specifies that standard errors for independently and identically distributed data be reported.

The standard derivative-based standard-error estimators cannot be used by `teffects psmatch`, because these matching estimators are not differentiable. The implemented method were derived by [Abadie and Imbens \(2006, 2011, 2012\)](#); see *Methods and formulas*.

As discussed in [Abadie and Imbens \(2008\)](#), bootstrap estimators do not provide reliable standard errors for the estimator implemented by `teffects psmatch`.

Reporting

`level(#)`; see [\[R\] estimation options](#).

`display_options`: `noomitted`, `vsquish`, `noemptycells`, `baselevels`, `allbaselevels`, `nofvlabel`, `fvwrap(#)`, `fvwrapon(style)`, `cformat(%fmt)`, `pformat(%fmt)`, `sformat(%fmt)`, and `no!stretch`; see [\[R\] estimation options](#).

Advanced

`caliper(#)` specifies the maximum distance at which two observations are a potential match. By default, all observations are potential matches regardless of how dissimilar they are.

In `teffects psmatch`, the distance is measured by the estimated propensity score. If an observation has no matches, `teffects psmatch` exits with an error.

`pstolerance(#)` specifies the tolerance used to check the overlap assumption. The default value is `pstolerance(1e-5)`. `teffects` will exit with an error if an observation has an estimated propensity score smaller than that specified by `pstolerance()`.

`osample(newvar)` specifies that indicator variable `newvar` be created to identify observations that violate the overlap assumption. Two checks are made to verify the assumption. The first ensures that the propensity scores are greater than `pstolerance(#)` and less than $1 - \text{pstolerance}(\#)$. The second ensures that each observation has at least `nneighbor(#)` matches in the opposite treatment group within the distance specified by `caliper(#)`.

The `vce(robust, nn(#))` option also requires at least `#` matches in the same treatment group within the distance specified by `caliper(#)`.

The average treatment effect on the treated, option `atet`, using `vce(iid)` requires only `nneighbor(#)` control group matches for the treated group.

`control(# | label)` specifies the level of `tvar` that is the control. The default is the first treatment level. You may specify the numeric level `#` (a nonnegative integer) or the label associated with the numeric level. `control()` and `tlevel()` may not specify the same treatment level.

`tlevel(# | label)` specifies the level of `tvar` that is the treatment for the statistic `atet`. The default is the second treatment level. You may specify the numeric level `#` (a nonnegative integer) or the label associated with the numeric level. `tlevel()` may only be specified with statistic `atet`. `tlevel()` and `control()` may not specify the same treatment level.

`generate(stub)` specifies that the observation numbers of the nearest neighbors be stored in the new variables `stub1`, `stub2`, `...`. This option is required if you wish to perform postestimation based on the matching results. The number of variables generated may be more than `nneighbors(#)` because of tied distances. These variables may not already exist.

The following option is available with `teffects psmatch` but is not shown in the dialog box:

`coeflegend`; see [R] [estimation options](#).

Remarks and examples

[stata.com](http://www.stata.com)

Propensity-score matching uses an average of the outcomes of similar subjects who get the other treatment level to impute the missing potential outcome for each subject. The average treatment effect (ATE) is computed by taking the average of the difference between the observed and potential outcomes for each subject. `teffects psmatch` determines how near subjects are to each other by using estimated treatment probabilities, known as propensity scores. This type of matching is known as propensity-score matching (PSM).

PSM does not need bias correction, because PSM matches on a single continuous covariate. In contrast, the nearest-neighbor matching estimator implemented in `teffects nnmatch` uses a bias-correction term when matching on more than one continuous covariate. In effect, the PSM estimator parameterizes the bias-correction term in the treatment probability model. See [TE] [teffects intro](#) or [TE] [teffects intro advanced](#) for more information about this estimator.

We will illustrate the use of `teffects psmatch` by using data from a study of the effect of a mother's smoking status during pregnancy (`mbsmoke`) on infant birthweight (`bweight`) as reported by Cattaneo (2010). This dataset also contains information about each mother's age (`mage`), education level (`medu`), marital status (`mmarried`), whether the first prenatal exam occurred in the first trimester (`prenatal1`), whether this baby was the mother's first birth (`fbaby`), and the father's age (`fage`).

► Example 1: Estimating the ATE

We begin by using `teffects psmatch` to estimate the average treatment effect of `mbsmoke` on `bweight`. We use a logistic model (the default) to predict each subject's propensity score, using covariates `mage`, `medu`, `mmarried`, and `fbaby`. Because the performance of PSM hinges upon how well we can predict the propensity scores, we will use factor-variable notation to include both linear and quadratic terms for `mage`, the only continuous variable in our model:

```
. use http://www.stata-press.com/data/r13/cattaneo2
(Excerpt from Cattaneo (2010) Journal of Econometrics 155: 138-154)
. teffects psmatch (bweight) (mbsmoke mmarried c.mage##c.mage fbaby medu)
Treatment-effects estimation      Number of obs      =      4642
Estimator      : propensity-score matching      Matches: requested =      1
Outcome model  : matching                      min =            1
Treatment model: logit                       max =            74
```

bweight	Coef.	AI Robust Std. Err.	z	P> z	[95% Conf. Interval]	
ATE						
mbsmoke (smoker vs nonsmoker)	-210.9683	32.021	-6.59	0.000	-273.7284	-148.2083

Smoking causes infants' birthweights to be reduced by an average of 211 grams.

◀

By default, `teffects psmatch` estimates the ATE by matching each subject to a single subject with the opposite treatment whose propensity score is closest. Sometimes, however, we may want to ensure that matching occurs only when the propensity scores of a subject and a match differ by less than a specified amount. To do that, we use the `caliper()` option. If a match within the distance specified in the `caliper()` option cannot be found, `teffects psmatch` exits.

▶ Example 2: Specifying the caliper

Here we reconsider the previous example, first specifying that we only want to consider a pair of observations a match if the absolute difference in the propensity scores is less than 0.03:

```
. teffects psmatch (bweight) (mbsmoke mmarried c.mage##c.mage fbaby medu),
> caliper(0.03)
no nearest-neighbor matches for observation 2209 within caliper 0.03; this
is not allowed
r(459);
```

The error arose because there is not a smoking mother whose propensity score is within 0.03 of the propensity score of the nonsmoking mother in observation 2209. If we instead raise the caliper to 0.10, we have matches for all subjects and therefore obtain the same results as in [example 1](#):

```
. teffects psmatch (bweight) (mbsmoke mmarried c.mage##c.mage fbaby medu),
> caliper(0.1)
Treatment-effects estimation      Number of obs      =      4642
Estimator      : propensity-score matching      Matches: requested =      1
Outcome model  : matching                      min =      1
Treatment model: logit                        max =      74
```

bweight	Coef.	AI Robust Std. Err.	z	P> z	[95% Conf. Interval]	
ATE						
mbsmoke (smoker vs nonsmoker)	-210.9683	32.021	-6.59	0.000	-273.7284	-148.2083

◀

□ Technical note

[Example 2](#) highlights that estimating the ATE requires finding matches for both the treated and control subjects. In contrast, estimating the ATET only requires finding matches for the treated subjects. Because subject 2209 is a control subject, we can estimate the ATET using `caliper(0.03)`. We must also specify `vce(iid)` because the default robust standard errors for the estimated ATET require viable matches for both treated subjects and control subjects. (This requirement comes from the nonparametric method derived by [Abadie and Imbens \[2012\]](#).)

```
. teffects psmatch (bweight) (mbsmoke mmarried c.mage##c.mage fbaby medu),
> atet vce(iid) caliper(0.03)
Treatment-effects estimation      Number of obs      =      4642
Estimator      : propensity-score matching      Matches: requested =      1
Outcome model  : matching                      min =      1
Treatment model: logit                        max =      74
```

bweight	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
ATET						
mbsmoke (smoker vs nonsmoker)	-236.7848	26.11698	-9.07	0.000	-287.9731	-185.5964

□

In the previous examples, each subject was matched to at least one other subject, which is the default behavior for `teffects psmatch`. However, we can request that `teffects psmatch` match each subject to multiple subjects with the opposite treatment level by specifying the `nneighbor()` option. Matching on more distant neighbors can reduce the variance of the estimator at a cost of an increase in bias.

Example 3

Now we request that `teffects psmatch` match a mother to four mothers in the opposite treatment group:

```
. teffects psmatch (bweight) (mbsmoke mmarried c.mage##c.mage fbaby medu),
> nneighbor(4)

Treatment-effects estimation           Number of obs       =      4642
Estimator       : propensity-score matching   Matches: requested =         4
Outcome model   : matching                   min =                 4
Treatment model: logit                       max =                 74
```

	Coef.	AI Robust Std. Err.	z	P> z	[95% Conf. Interval]	
<hr/>						
ATE						
mbsmoke (smoker vs nonsmoker)	-224.006	29.88627	-7.50	0.000	-282.582	-165.43

◀

Stored results

`teffects psmatch` stores the following in `e()`:

Scalars

<code>e(N)</code>	number of observations
<code>e(nj)</code>	number of observations for treatment level j
<code>e(k_levels)</code>	number of levels in treatment variable
<code>e(caliper)</code>	maximum distance between matches
<code>e(treated)</code>	level of treatment variable defined as treated
<code>e(control)</code>	level of treatment variable defined as control
<code>e(k_nneighbor)</code>	requested number of matches
<code>e(k_nnmin)</code>	minimum number of matches
<code>e(k_nnmax)</code>	maximum number of matches
<code>e(k_robust)</code>	matches for robust VCE

Macros

<code>e(cmd)</code>	<code>teffects</code>
<code>e(cmdline)</code>	command as typed
<code>e(depvar)</code>	name of outcome variable
<code>e(tvar)</code>	name of treatment variable
<code>e(subcmd)</code>	<code>psmatch</code>
<code>e(tmodel)</code>	logit, probit, or hetprobit
<code>e(stat)</code>	statistic estimated, ate or atet
<code>e(wtype)</code>	weight type
<code>e(wexp)</code>	weight expression
<code>e(title)</code>	title in estimation output
<code>e(tlevels)</code>	levels of treatment variable
<code>e(psvarlist)</code>	variables in propensity-score model
<code>e(hvarlist)</code>	variables for variance, only if <code>hetprobit</code>
<code>e(vce)</code>	<code>vcetype</code> specified in <code>vce()</code>
<code>e(vcetype)</code>	title used to label Std. Err.
<code>e(datasignature)</code>	the checksum
<code>e(datasignaturevars)</code>	variables used in calculation of checksum
<code>e(properties)</code>	<code>b V</code>
<code>e(estat_cmd)</code>	program used to implement <code>estat</code>
<code>e(predict)</code>	program used to implement <code>predict</code>
<code>e(marginsnotok)</code>	predictions disallowed by <code>margins</code>

Matrices

<code>e(b)</code>	coefficient vector
<code>e(V)</code>	variance-covariance matrix of the estimators
<code>e(bps)</code>	coefficient vector from propensity-score model
<code>e(Vps)</code>	variance-covariance matrix of the estimators from propensity-score model

Functions

<code>e(sample)</code>	marks estimation sample
------------------------	-------------------------

Methods and formulas

The methods and formulas used by `teffects psmatch` are documented in the *Methods and formulas* of [TE] `teffects nnmatch`.

References

- Abadie, A., and G. W. Imbens. 2006. Large sample properties of matching estimators for average treatment effects. *Econometrica* 74: 235–267.
- . 2008. On the failure of the bootstrap for matching estimators. *Econometrica* 76: 1537–1557.
- . 2011. Bias-corrected matching estimators for average treatment effects. *Journal of Business and Economic Statistics* 29: 1–11.
- . 2012. Matching on the estimated propensity score. Harvard University and National Bureau of Economic Research. <http://www.hks.harvard.edu/fs/aabadie/pscore.pdf>.
- Cattaneo, M. D. 2010. Efficient semiparametric estimation of multi-valued treatment effects under ignorability. *Journal of Econometrics* 155: 138–154.

Also see

- [TE] `teffects postestimation` — Postestimation tools for `teffects`
- [TE] `teffects` — Treatment-effects estimation for observational data
- [TE] `teffects nnmatch` — Nearest-neighbor matching
- [R] `hetprobit` — Heteroskedastic probit model
- [R] `logit` — Logistic regression, reporting coefficients
- [R] `mlogit` — Multinomial (polytomous) logistic regression
- [R] `probit` — Probit regression
- [U] **20 Estimation and postestimation commands**